THE IMPORTANCE OF THE SIMULATION EXPECTATION AS A GOODNESS OF FIT DIAGNOSTIC FOR CATEGORICAL POPULATION PHARMACODYNAMIC MODELS

Marc R. Gastonguay, Jeffrey T. Hane
Metrum Research Group LLC, Avon, CT 06001

Which Model is the Base Model?

MODEL A

Observed



Predicted



MODEL B

Observed



Predicted



Answer: Both are the Base Model

MODEL A



MODEL B



Introduction

- Most goodness-of-fit (GOF) diagnostics for analysis of categorical population (POP) pharmacodynamic (PD) data using nonlinear mixed effects models (NLMEM) involve comparisons of observed and predicted probabilities under the conditional expectation (CE) from the model.

- The simulation expectation (SE) has also been proposed as a GOF diagnostic.

CE vs. SE

- **Conditional Expectation (CE)**: predictions based on conditional estimates of individual random effects (ETAs) given population model and individual data

- **Simulation Expectation (SE)**: predictions based on randomly simulated ETAs from estimated population inter-individual variance distribution and population model

Objective

- To compare the performance of CE and SE GOF diagnostics for ordered-categorical POP PD models

Methods

- Simulations of a hypothetical POP PD database (400 individuals, 930 total PD observations) for an ordered categorical response were conducted under the assumptions of true drug and covariate effects.

- Simulated data were subsequently analyzed using both a null effect (base model) and a full covariate model with NLMEM. For each simulation, GOF diagnostic plots were created using the CE and the SE.

Estimation Methods

- Population PD Modeling via Nonlinear Mixed-Effects (NONMEM software with Laplacian estimation method)
- Proportional odds model for ordinal categorical data implemented in mixed-effects modeling context
- Cumulative probabilities across score levels were described with cumulative logit functions.
- M-1 cumulative logits for an M-category response

Cumulative Logit Model for Ordered Categorical Response

$$g\{P(Y_t \le m \mid \eta)\} = \text{logit}(p) = f_m + \eta$$

$$\text{logit}(p) = \text{Log}[p/(1-p)]$$

$$p = \frac{\exp^{f_m + \eta}}{1 + \exp^{f_m + \eta}}$$

where :

$$\eta \sim N(0, \omega^2)$$

$$f_m = \sum_{L=1}^{m} \alpha_{L_i} + f(\text{AUC}, \text{Treatment}, \text{covariates}, \text{etc.})$$

Results

- The CE diagnostics consistently revealed an overly-optimistic GOF (Figure 1).

- When the base model was used for estimation, bias in GOF description was up to +300% for the lowest categorical score.

- The SE resulted in a more realistic depiction of model GOF.

Figure 1: Base Model Diagnostics



Figure 2: Covariate Effects-Base Model



Figure 3: Final Model Diagnostics



Figure 4: Covariate Effects-Final Model



Figure 5: Random Effects Distributions





Conclusions

- For ordered categorical POP PD models, the SE is a more reliable and accurate descriptor of GOF than the CE; the CE should not be used for this purpose.
- The CE is useful for identifying covariate-related trends in random effects (Figures 2 & 4) and for checking assumptions about distributions of random effects (Figure 5).
- Similar results are expected with other categorical or odd-type POP PD models.

NMTRAN Control Stream: Simulation

```
$PROB Outcome Score PD Model: FULL
      MODEL SIM
$INPUT C DV VISD ID TRT SEX WT AGE
      BLIN BIO AUC OBS
$DATA PD5.CSV IGNORE=C
$PRED
; ORDINAL CATEGORICAL RESPONSE
; CUMULATIVE ODDS RATIO MODEL
; (SEE AGRESTI, A. CATEGORICAL
; DATA ANALYSIS. 1990 WILEY. PP
; 322-331.)
; SCALE = 0, 1, 2, 3, 4 (OUTCOME
; SCORE RANGING FROM GOOD [0] TO
; WORST ; POSSIBLE [4])
; BENEFICIAL DRUG EFFECT
; INCREASES PROBABILITY OF LOW
; SCORES
; cutpoint values (alpha[j] in Agresti
;  notation)
 AL0 = THETA(1)    ;SCORE <=0
 AL1 = AL0+THETA(2) ;SCORE <=1
 AL2 = AL1+THETA(3) ;SCORE <=2
 AL3 = AL2+THETA(4) ;SCORE <=3

; CONCENTRATION-RESPONSE
;  relationship (beta*x in Agresti notation)
; TRT 1=DRUG, 0=PLACEBO
; BLIN 0 =LOW, 1=HIGH BASELINE
; SYMPTOMS
; BIO IS BIOMARKER AUC
; AGE (YEARS)
; AUC = DRUG AUC
E= THETA(5)*BLIN + THETA(6)*BIO +
     THETA(7)*AGE +THETA(8)*AUC
```

NMTRAN Control Stream: Simulation

```
; logits for cumulative probabilities
; X IS PREDICTOR VARIABLE
; J = 0,1,2,3 AND IS DENOTED BY [J]
 LGT0 = AL0+E+ETA(1) ;LGT[0](X)
 LGT1 = AL1+E+ETA(1) ;LGT[1](X)
 LGT2 = AL2+E+ETA(1) ;LGT[2](X)
 LGT3 = AL3+E+ETA(1) ;LGT[3](X)

; exponentiate logit
C0 = EXP(LGT0)
C1 = EXP(LGT1)
C2 = EXP(LGT2)
C3 = EXP(LGT3)

; CUMULATIVE PROBABILITIES
 P0 = C0/(1+C0) ;P(Y<=0|X)
 P1 = C1/(1+C1) ;P(Y<=1|X)
 P2 = C2/(1+C2) ;P(Y<=2|X)
 P3 = C3/(1+C3) ;P(Y<=3|X)

; LIKELIHOOD FOR CATEGORY=J
 Y0=P0
 Y1=P1-P0
 Y2=P2-P1
 Y3=P3-P2
 Y4=1-P3

; SIMULATION EXPECTATION
IF(ICALL.EQ.4) THEN ;simulation
  CALL RANDOM (2,R)
  IF(R.LE.P0) DV=0
  IF(R.GT.P0.AND.R.LE.P1) DV=1
  IF(R.GT.P1.AND.R.LE.P2) DV=2
  IF(R.GT.P2.AND.R.LE.P3) DV=3
  IF(R.GT.P3) DV=4
  X2=R
ENDIF
```

NMTRAN Control Stream: Simulation

```
IND0=0
IND1=0
IND2=0
IND3=0
IND4=0
IF (DV.EQ.0) IND0=1
IF (DV.EQ.1) IND1=1
IF (DV.EQ.2) IND2=1
IF (DV.EQ.3) IND3=1
IF (DV.EQ.4) IND4=1

IF(BIO.LE.50) BIOM=1
IF(BIO.GT.50.AND.BIO.LE.100) BIOM=2
IF(BIO.GT.100.AND.BIO.LE.150) BIOM=3
IF(BIO.GT.150) BIOM=4
```

```
IF(AUC.EQ.0) ACAT=1
 IF(AUC.GT.0.AND.AUC.LE.20000) ACAT=2
 IF(AUC.GT.20000.AND.AUC.LE.40000)
     ACAT=3
 IF(AUC.GT.40000.AND.AUC.LE.60000)
     ACAT=4
 IF(AUC.GT.60000.AND.AUC.LE.80000)
     ACAT=5
 IF(AUC.GT.80000) ACAT=6

Y=Y0*IND0+Y1*IND1+Y2*IND2+Y3*IND3+
     Y4*IND4

$MSFI 007.msf ; from estimation run
$SIM (95465) (546 UNIFORM) ONLY
     SUB=100 NOPRED
$TABLE NOPRINT NOHEADER
     FILE=008.TAB
 ID ETA(1) P0 P1 P2 P3 Y0 Y1 Y2 Y3 Y4 VISD
     BIO TRT BLIN AGE AUC BIOM ACAT
     OBS
```

References

Agresti, A. *Categorical Data Analysis*. John Wiley & Sons Inc., New York, 1990.

Kalbfleisch J.G., *Probability and Statistical Inference, Volume 1: Probability.* 2nd Edition. Springer-Verlag, New York, 1985.

Sheiner, L.B. A New Approach to the Analysis of Analgesic Drug Trials, Illustrated with Bromfenac Data. *Clin. Pharmacol. Ther.* 1994; 56:309-322.

Ware L.H., and Lipsitz, S. Issues in the Analysis of Repeated Categorical Outcomes. *Statistics in Medicine.* Vol. 7: 95-107, 1988.

For copies of this poster, please visit www.metrumrg.com